

Auto-populating Microsoft PowerPoint Presentations So You Don't Have To

Kaitie Lawson and Brett Jepson, Rho, Inc.

ABSTRACT

In the field of clinical research, Data Safety Monitoring Board (DSMB) members are increasingly requesting high-level overview presentations to assess the current status of a clinical trial in lieu of lengthy binders. When multiple clinical trials report to the same DSMB, a certain level of standardization across presentations is desirable to facilitate ease in understanding and familiarity with the presentation format for the DSMB members. Manually entering text and data into a presentation is tedious and is subject to transcription errors. By utilizing ODS PowerPoint, ODS Text, ODS List, ODS Layout and a series of other familiar SAS procedures, we have created a suite of SAS macros that automatically populate presentations, therefore eliminating the need to manually enter data into the presentation slides. These SAS macros allow for presentations conforming to a general template to be created in a standardized manner which saves time, ensures data accuracy and integrity, and provides continuity in data presentation to DSMB members, while at the same time providing a fair amount of customization. We will share basics of how to create your own macros to auto populate PowerPoint presentations, as well as some tips and tricks we have learned along the way.

INTRODUCTION

Data Safety Monitoring Board (DSMB) members are increasingly requesting high-level overview presentations to assess the current status of a trial in lieu of reviewing lengthy binders. The need for succinct and informative slides, tables, and graphics are imperative so DSMB members can obtain the information that they need to make decisions about how (or if) a trial should continue. Traditionally, data and information from summary tables and displays would need to be entered into a presentation manually. However, manually entering text and data into a presentation is tedious and is subject to transcription errors. Having a template auto populated by utilizing ODS PowerPoint, ODS Text, ODS List, and a series of other familiar SAS procedures circumvents the potential for human error and saves time. This paper will show examples of how one program can be created to produce all the slides necessary for a DSMB presentation by automatically populating an existing template.

ODS POWERPOINT OVERVIEW

The ODS PowerPoint destination has been available in SAS 9.4 since 2013 which allows users to format SAS procedure output as PowerPoint slides. Accessing the destination for PowerPoint is similar to accessing other ODS destinations by simply specifying the POWERPOINT keyword in the ODS statement in conjunction with the FILE= option:

```
ODS POWERPOINT FILE="Example.pptx";
```

```
<Additional SAS Code>
```

```
ODS POWERPOINT CLOSE;
```

There are essentially three types of data we want to incorporate into PowerPoint slides for DSMB reports: text, tables and figures. We will highlight how to produce these slides using the REPORT, SGPLOT, ODS TEXT and ODSLST procedures in conjunction with the ODS POWERPOINT destination. After covering these basic procedures, we will show how an entire DSMB PowerPoint presentation can be created from one single program and how to create SAS macros so that SAS programs generating DSMB presentations can accommodate multiple studies.

TEXT SLIDES USING THE ODS TEXT AND ODSLST PROCEDURES

All PowerPoint slides contain some form of text. When we refer to 'Text Slide', we are mainly referring to title slides or slides that contain bulleted/summary information without any displays (i.e., tables or figures). For example, Figure 1 and Figure 2 display typical text slides that you may see at the beginning of any DSMB report.

Figure 1 is a title slide, and Figure 2 is a bulleted slide describing the study design.

Title of the DSMB Presentation

Protocol Number: Study-01
Sponsor: Pharma Company

Report Prepared by:
My CRO, Inc. (My City, USA)
Data Cutoff Date: February 6, 2018
DSMB Meeting Date: April 9, 2018
Open Meeting Presentation

Figure 1. Example of a Title Slide

Study Design

- Phase 3, 5-site, 2-arm, double-blind, placebo-controlled, randomized trial in participants with allergic rhinitis
 - Arm 1: Sublingual-Immunotherapy
 - Arm 2: Placebo
- Enrollment goal: 100 participants randomized
- Primary Endpoint: Change in FEV1 % predicted from baseline

Figure 2. Example of a Bulleted List Slide

It might not be intuitive that creating text slides in SAS would save time; admittedly, creating text slides is fairly simple in PowerPoint. However, when we create text slides in ODS POWERPOINT, we allow for the possibility to create the entire presentation within one program. Otherwise, text slides would either need to be hand entered after using ODS POWERPOINT to create other slides with tables or graphics, or copied and pasted from a previous presentation. Both of those methods introduce the potential for typographical errors. In addition, since DSMB meetings are often repeated at regular intervals (e.g. every 6 months, once a year, etc.), including code to create all slides of a subsequent presentation within a single SAS program allows for easy reruns with little to no modification.

EXAMPLE 1: TITLE SLIDE USING ODSTEXT

The code to produce the slide in Figure 1, which uses the ODSTEXT procedure, is below:

```
ODS ESCAPECHAR='~';  
ODS POWERPOINT FILE="TITLE_SLIDE.pptx" NOGTITLE NOGFOOTNOTE;  
PROC ODSTEXT;  
  P " " / STYLE={FONTSIZE=30};  
  P "Title of the DSMB Presentation" / STYLE={FONTSIZE=20 JUST=C  
FONTWEIGHT=BOLD};
```

```

P " " / STYLE={FONTSIZE=30};
P "Protocol Number: Study-01~n Sponsor: Pharma Company" /
STYLE={FONTSIZE=12 JUST=C};
P "Report Prepared by:~n My CRO, Inc. (My City, USA)" / STYLE ={FONTSIZE=8
JUST=C};
P "Data Cutoff Date: February, 6 2018~n DSMB Meeting Date: April 9, 2018~n
Open Meeting Presentation" / STYLE ={FONTSIZE=8 JUST=C};
RUN;
ODS POWERPOINT CLOSE;

```

Each “P” statement creates a new text block. There are a variety of options that can be included as STYLE elements to change the orientation and font of text. For this slide, we decided that all the text should be center justified (JUST=C) while we have used varying font sizes (FONTSIZE) and font weights (FONTWEIGHT). Two blank paragraphs (P statements) are included to add spacing above and below the title.

The “~n” code creates a hard return and moves all subsequent text to the next line. The escape character, which is defined in the ODS ESCAPECHAR statement before the ODS POWERPOINT statement, was chosen to be ‘~’ in our code, but could be set to any acceptable escape character allowable in SAS. This method allows for text to be grouped in the same text box in the PowerPoint slide, while including text in a new P statement creates a new text box. The amount of space between text boxes will generally be larger than the space between lines within a text box.

Below is an example of how to macroize the preceding code:

```

ODS ESCAPECHAR='~';
%MACRO TITLESIDE
(PRESENTATION_TITLE=, PROTOCOL_NUMBER=, SPONSOR=, CUTOFF_DATE=, DSMB_DATE=);
ODS POWERPOINT FILE="TITLE_SLIDE.pptx" NOGTITLE NOGFOOTNOTE;
PROC ODSTEXT;
P " " / STYLE={FONTSIZE=30};
P "&PRESENTATION_TITLE." / STYLE={FONTSIZE=20 JUST=C FONTWEIGHT=BOLD};
P " " / STYLE={FONTSIZE=30};
P "Protocol Number: &PROTOCOL_NUMBER.~n Sponsor: &SPONSOR." /
STYLE={FONTSIZE=12 JUST=C};
P "Report Prepared by:~n My CRO, Inc. (My City, USA)" / STYLE ={FONTSIZE=8
JUST=C};
P "Data Cutoff Date: &CUTOFF_DATE.~n DSMB Meeting Date: &DSMB_DATE.~n Open
Meeting Presentation" / STYLE ={FONTSIZE=8 JUST=C};
RUN;
ODS POWERPOINT CLOSE;
%MEND TITLESIDE;

%TITLESIDE (PRESENTATION_TITLE= Title of the DSMB Presentation
, PROTOCOL_NUMBER= Study-01
, SPONSOR= Pharma Company
, CUTOFF_DATE= %STR(February 6, 2018)
, DSMB_DATE= %STR(April 9, 2018))

```

The %STR macro function is used to treat the value as text only. For example, when a value is a date with a comma, as in the last two macro variables (CUTOFF_DATE and DSMB_DATE), %STR masks the commas when resolving the macro call. Without using the %STR function, the macro will assume that the clause after the comma (in this case, the year) is a new macro variable definition and will give a WARNING statement concerning positional parameters. Other macro functions that are similar to %STR can also be used depending on which special characters are needed to be masked (e.g. %BQUOTE and %NRBQUOTE).

EXAMPLE 2: STUDY DESIGN SLIDE USING ODSLST

The code to produce the slide in Figure 2, which uses the ODSLIS T procedure, is below:

```
ODS POWERPOINT FILE = "DESIGN_SLIDE.PPTX";
PROC ODSLIS T;
  ITEM;
    P "Phase 3, 5-site, 2-arm, double-blind, placebo-controlled, randomized
trial in participants with allergic rhinitis" || " " /STYLE={FONTSIZE=12
LISTSTYLETYPE= "disc"};
    LIST / STYLE={FONTSIZE=10.5 LISTSTYLETYPE="circle"};
      ITEM "Arm 1: Sublingual-Immunotherapy";
      ITEM "Arm 2: Placebo";
  END;
END;
ITEM "Enrollment goal: 100 participants randomized" || " "
/STYLE={FONTSIZE=12 LISTSTYLETYPE="disc"};
ITEM "Primary Endpoint: Change in FEV1 % predicted from baseline"
/STYLE={FONTSIZE=12 LISTSTYLETYPE="disc"};
RUN;
ODS POWERPOINT CLOSE;
```

The ODSLIS T procedure uses ITEM blocks to create lists, which can be customized using LISTSTYLETYPE in its appearance (numbered, bulleted, etc.). In our example, we have specified LISTSTYLETYPE="disc" which creates the filled in circle or 'disc' for the bullet and sub bullet points. In addition, sub-lists can be created within an ITEM. This is done by including a set of ITEM statements under a LIST statement.

If we create macro variables to generate the key elements of the design slide (i.e. Design, Arms, Enrollment Goal, and Primary Endpoint), we could have one program that would create this exact same slide for any number of studies which in turn provides efficiency and standardization across all presentations made during DSMB meetings. The following example provides the macro code and two different macro calls for two completely different studies. Figure 3 shows the two different design slides that are produced from the two individual macro calls.




Figure 3. Study Design Slides Produced from Two Separate Macro Calls

The following macro and subsequent macro calls create both of the slides presented in Figure 3:

```

%MACRO STUDYDESIGN (DESIGN=, ARMS=, ENROLLGOAL=, PRIMARY=, FOOTNOTE=);
ODS POWERPOINT FILE = "DESIGN_SLIDE.PPTX";
TITLE "Study Design";
FOOTNOTE;
%IF &FOOTNOTE. ^= %THEN FOOTNOTE J=L "&FOOTNOTE";;
ODS ESCAPECHAR='~';

/*Create a macro variable that counts the number of arms and primary
endpoints*/
DATA _NULL_;
  ARMS = "&ARMS.";
  NUMARMS = COUNT(ARMS, '|') + 1;
  CALL SYMPUTX("NUMARMS", NUMARMS);
  PRIMARY = "&PRIMARY.";
  NUMPRIMARY = COUNT(PRIMARY, '|') + 1;
  CALL SYMPUTX("NUMPRIMARY", NUMPRIMARY);
RUN;

PROC ODSLIST;
  ITEM;
    P "&DESIGN." %IF &ARMS. = %THEN || "~n "; / STYLE={FONTSIZE=12
LISTSTYLETYPE="disc"};
    %IF &ARMS. ^= %THEN %DO;
      LIST / STYLE={FONTSIZE=10.5 LISTSTYLETYPE="circle"};
      %DO I = 1 %TO &NUMARMS.;
        ITEM "Arm &I.: %SCAN(&ARMS., &I., |)"
          %IF &I. = &NUMARMS. %THEN || "~n ";;
      %END;
    END;
  %END;
END;
  ITEM "Enrollment goal: &ENROLLGOAL.~n" || " " /STYLE={FONTSIZE=12
LISTSTYLETYPE="disc"};
  %IF &NUMPRIMARY. = 1 %THEN %DO;
    ITEM "Primary endpoint: &PRIMARY." /STYLE={FONTSIZE=12
LISTSTYLETYPE="disc"};
  %END;
  %ELSE %DO;
    ITEM;
      P "Primary endpoints:" / STYLE={FONTSIZE=12 LISTSTYLETYPE="disc"};
      LIST / STYLE={FONTSIZE=10.5 LISTSTYLETYPE="circle"};
      %DO J = 1 %TO &NUMPRIMARY.;
        ITEM "%SCAN(&PRIMARY., &J., |)";
      %END;
    END;
  %END;

RUN;
ODS POWERPOINT CLOSE;
%MEND STUDYDESIGN;

%STUDYDESIGN(DESIGN= %STR(Phase 2, 3 site, 2 arm, double-blind, placebo-
controlled, randomized trial in adults aged 18-55 years with Lupus Nephritis)
, ARMS= %STR(A|B)

```

```

,ENROLLGOAL= 100 Randomized
,PRIMARY= Proportion of subjects who achieve a complete response at 24
weeks
);

%STUDYDESIGN(DESIGN= %STR(Phase 3, 1 site, 3 arm, double-blind, double-dummy,
placebo-controlled, randomized trial in adults aged 18-55 with Seasonal
Allergic Rhinitis)
,ARMS= %STR(Subcutaneous Immunotherapy + Sublingual Placebo|Sublingual
Immunotherapy + Subcutaneous Placebo|Double Placebo)
,ENROLLGOAL= %STR(300 randomized in a 1:1:1 fashion)
,PRIMARY= %STR(Total Nasal Symptom Score Area Under the Curve in response
to Nasal Allergen Challenge measured at 3 Years|Peak Nasal Inspiratory Flow
Area Under in response to Nasal Allergen Challenge the Curve measured at 3
Years));

```

The macro code that we have designed is flexible enough to create a Design Slide for a study with any number of arms and primary endpoints. Having these options built into the code allows for a fair amount of customization and flexibility. As you can see, the code used to create these slides is contained within the macro called %STUDYDESIGN. We have also developed a macro called %STUDYDESIGN_OBSERVATIONAL (utilizing similar coding logic) with its own set of parameters in case the study that needs to be reported to the DSMB is observational as opposed to an interventional trial.

TEXT SLIDE HELPFUL HINT #1

ODS LAYOUT may also be helpful in customizing slides to fit your needs since it allows multiple lists, tables or figures to be created and displayed on the same slide. We found ODS LAYOUT useful in creating a slide containing an outline of the presentation, with two text boxes side by side. The left side will contain the slide name, while the right side will contain the corresponding slide number. ODS LAYOUT code is structured like this:

```

ODS LAYOUT GRIDDED COLUMNS=2;

ODS REGION WIDTH=5IN;
<CODE TO CREATE LEFT COLUMN>

ODS REGION WIDTH=3IN;
<CODE TO CREATE RIGHT COLUMN>

ODS LAYOUT END;

```

TEXT SLIDE HELPFUL HINT #2

For lists that could contain a varying number of items depending on the particular report, we found it useful to include a macro variable that contains all list items delimited by a pipe (|). By doing this, it is possible to include as many items as necessary for that particular report. We chose the pipe as a delimiter because it was unlikely to appear in any other context. Only list items will appear as bulleted items in the slide. An example of a macro that demonstrates this idea is shown here:

```

%MACRO APPENDIX(LIST=);
ODS POWERPOINT FILE = "APPENDIX.PPTX";
ODS ESCAPECHAR='~';
/*Create a macro variable that counts the number of list items*/
DATA _NULL_;
  ITEMS = "&LIST.";

```

```

NUMITEMS = COUNT (ITEMS, '|') + 1;
CALL SYMPUTX ("NUMITEMS", NUMITEMS);
RUN;

/*Create a macro variable that counts the number of list items*/
TITLE "APPENDICES";
PROC ODSLIST;
  ITEM;
  LIST / STYLE={FONTSIZE=16 LISTSTYLETYPE="UPPER_ALPHA"};
  %DO I = 1 %TO &NUMITEMS.;
    ITEM "%SCAN (&LIST., &I., |)";
  %END;
END;
END;
RUN;
ODS POWERPOINT CLOSE;
%MEND APPENDIX;

%APPENDIX (LIST=%STR (PROTOCOL|AE LISTING) );

```

TABLE SLIDES USING THE REPORT PROCEDURE

Table slides refer to PowerPoint slides that include summary data presented in a tabular format. These are the slides where the largest potential for transcription errors could occur and therefore provide the best opportunity to increase efficiencies. Figure 4 displays a table slide summarizing protocol deviations.

Protocol Deviations

Deviation	Deviations: n (%)	
	Since Last Report: N = 20	To Date: N = 50
Informed Consent	4 (20%)	8 (16%)
Missed Safety, Efficacy or Endpoint Assessments	7 (35%)	12 (24%)
Reporting of SAEs	1 (5%)	4 (8%)
Study treatment administration	5 (25%)	14 (28%)
Use of Prohibited Medications	1 (5%)	5 (10%)
Other: Violation of visit window	2 (10%)	7 (14%)

Figure 4. Protocol Deviations Table

EXAMPLE 3: PROTOCOL DEVIATIONS SUMMARY SLIDE USING PROC REPORT

This table summarizes protocol deviations that subjects have experienced during the entire study (as seen in the 'To Date' column) as well as protocol deviations that have occurred since the last DSMB report (the 'Since Last Report' column). At study start up, each study team defines a set of reasons for documenting protocol deviations. However, more often than not, additional deviations occur that are not

in the pre-defined list. The macro code that we have developed is flexible enough to summarize these 'Other' deviations as they are also important to report to the DSMB.

The code used to make the Protocol Deviation slide is listed below:

```
ODS POWERPOINT FILE = "DEVIATION SLIDE.PPTX";
TITLE "Protocol Deviations";
PROC REPORT DATA=SUMMARY_DATA NOWD SPLIT = "~" HEADSKIP CENTER

STYLE(REPORT) = [HEIGHT=6IN WIDTH=9.25IN]
STYLE(HEADER)=[BACKGROUND=BIGB FONTWEIGHT=BOLD FOREGROUND=WHITE];

COLUMN LINE COL1 ("Deviations: n (%)") COL2 COL3);

DEFINE LINE / ORDER NOPRINT;

DEFINE COL1 / DISPLAY LEFT FLOW "~Deviation" STYLE=[JUST=LEFT CELLWIDTH=7IN]
STYLE(COLUMN)=[ASIS=ON] STYLE(HEADER) = [JUST=LEFT FONTWEIGHT=BOLD
FONTSIZE=5.8 FOREGROUND=WHITE BACKGROUND=BIGB];

DEFINE COL2 / DISPLAY LEFT FLOW "Since Last Report:~N = 25"
STYLE=[JUST=CENTER CELLWIDTH=2IN] COLOR=WHITE STYLE(HEADER) = [JUST=CENTER
FONTWEIGHT=BOLD FONTSIZE=5.8 FOREGROUND=WHITE BACKGROUND=BIGB];

DEFINE COL3 / DISPLAY LEFT FLOW "To Date:~N = 38" STYLE=[JUST=CENTER
CELLWIDTH=2IN] COLOR=WHITE STYLE(HEADER) = [JUST=CENTER FONTWEIGHT=BOLD
FONTSIZE=5.8 FOREGROUND=WHITE BACKGROUND=BIGB];

RUN;
ODS POWERPOINT CLOSE;
```

Furthermore, the DSMB might also be interested in seeing how the protocol deviations are distributed across each site in a multi-site study. An example of this type of table is shown in Figure 5.

Protocol Deviations by Site

Site	Deviations: n (%)	
	Since Last Report: N = 20	To Date: N = 50
Atlanta	0 (0%)	2 (4%)
Boston	1 (5%)	5 (10%)
Dallas	12 (60%)	21 (42%)
Denver	4 (20%)	11 (22%)
New York	1 (5%)	4 (8%)
Washington, DC	2 (10%)	7 (14%)

Figure 5. Protocol Deviations Table by Site

The code to create this table is similar to the code used to create the table in Figure 4. Because of these similarities, creating these two tables within one macro eliminates the amount of code needed to create them resulting in more efficiency and fewer opportunities to make mistakes. Although single-site studies would not need a by-site table, a macro with the capabilities to allow for a user to decide if the 'by site' table is included in a presentation meets the needs of all studies.

The following code structure shows how using simple macro code within a macro can make the Protocol Deviations by Site table optional:

```
ODS POWERPOINT FILE = "DEVIATION SLIDE.PPTX";
TITLE "Protocol Deviations";
<CODE TO CREATE A PROTOCOL DEVIATIONS TABLE>
%IF &BYSITE. = Y %THEN %DO;
    TITLE "Protocol Deviations by Site";
    <CODE TO CREATE A PROTOCOL DEVIATIONS BY SITE TABLE>
%END
ODS POWERPOINT CLOSE;
```

The BYSITE macro variable is a variable that would need to be included in the macro variables. When set to Y in the macro call, the by-site table slide is created. Otherwise, the by-site table slide is not created.

TABLE SLIDE HELPFUL HINT #1

It is helpful if datasets are standardized across the studies that will be using the same DSMB presentation template. Standardized datasets allow for the reduction in dataset manipulation and fewer macro variables to be defined which allows for default values to be set when developing the macro code. Datasets that usually require little customization (e.g., adverse events and protocol deviations) often make up the bulk of the information that DSMBs are interested in and are therefore excellent candidates for standardization between studies.

TABLE SLIDE HELPFUL HINT #2

When reporting categorical variables in tables where there is an 'Other, specify' option, an easy way to distinguish the 'Other, specify' items in the table is to create a shell of all options in the pre-defined list, while adding a prefix of 'Other:' to deviations not in the pre-defined list. Here is some code that performs this task, which would be inserted in the middle of a macro:

```
/*Create a macro variable that counts the number of deviation types*/
DATA _NULL_;
    TYPES = "&TYPESLIST.";
    NUMTYPES = COUNT(TYPES, '|') + 1;
    CALL SYMPUTX("NUMTYPES", NUMTYPES);
RUN;

/*Create a shell with 1 record per deviation type*/
DATA SHELL;
    FORMAT COL1_ $132.;
    TYPES = "&TYPESLIST.";
    %DO I = 1 %TO &NUMTYPES.;
        COL1_ = SCAN(TYPES, &I., '|');
        SORT = &I.;
    OUTPUT;
    %END;
RUN;
```

```

PROC SORT DATA = SHELL;
  BY COL1_;
RUN;

/*Merge deviation dataset with the shell and append "Other:" to anything */
/*that is not in the list of case report form-specified deviations. Also, */
/*sort the "Other" types last so they appear at the end of the list in the*/
/*table*/
DATA DEVIATIONS;
  LENGTH COL1 $132;
  MERGE DEVIATION_DATASET SHELL (IN=A);
  BY COL1_;
  IF A THEN COL1 = COL1_;
  ELSE DO;
    COL1 = "Other: " || COL1_;
    SORT = 99;
  END;
RUN;

```

This method ensures that no deviations are left out inadvertently, as they are classified with the ‘Other’ reason for deviations. The corresponding macro call would use a pipe (|) to separate the reasons for deviation and looks something like this:

```

%DEVIATION_TABLE (... , TYPELIST=%STR(REASON FOR DEVIATION 1|REASON FOR DEVIATION
2|...));

```

TABLE SLIDE HELPFUL HINT #3

For an initial DSMB meeting, the column for “Since Last Report” won’t be needed as there have been no previous reports; therefore, it is prudent to include an option that will include or exclude this column dependent upon the number of times a study has been presented to a DSMB. Similar coding logic as the code to include a by-site table (see Example 3) can be used to include or exclude a column in the REPORT procedure. It is necessary to derive a variable that determines which data have occurred since the last report which can be done by simply including a date that will act as a cut off in your macro coding.

FIGURE SLIDES USING THE SGPLOT PROCEDURE

Figure slides have the ability to convey complex, multifaceted information about the status of a study in a succinct manor. One of the key aspects of a study that the DSMB likes to track is enrollment over time. If study enrollment is too slow, this might mean that the enrollment criteria are too strict. A slow enrollment will also have an effect on the budget for the study. If study enrollment seems to be taking longer than expected, the DSMB might suggest protocol modifications, such as adjusting the inclusion and exclusion criteria, recruiting more sites for participation, etc. An example of a Figure Slide that displays enrollment over time is displayed in Figure 6.

Enrollment By Study Month




Figure 6. Enrollment over Time Figure

The code used to create this slide is below:

```
ODS POWERPOINT FILE = 'ENROLLMENT SLIDE.PPTX';
PROC SGLOT DATA = PLOT NOAUTOLEGEND;
    SERIES X=PRO Y=SUBSP / LINEATTRS=(COLOR=BLACK) LEGENDLABEL="Expected"
NAME="e";
    STEP X=MONTH Y=SUBS / LINEATTRS=(COLOR=BLUE) CURVELABEL = "N= 205 at Month
12" CURVELABELATTRS=(SIZE=9PT) LEGENDLABEL="Actual" NAME="a";

    XAXIS LABEL = "Study Month" OFFSETMAX=0.25;
    YAXIS LABEL = "Number Randomized" OFFSETMAX=.05;

    KEYLEGEND "e" "a" / NOBORDER LOCATION=INSIDE POSITION=TOPLEFT;
    TITLE "Enrollment By Study Month" BOLD H=35PT;
RUN;
ODS POWERPOINT CLOSE;
```

Just like with the text slides, it might not be intuitive that using ODS POWERPOINT and PROC SGLOT would save time when creating figure slides for a presentation. However, if we create macro variables to populate a few key elements of this slide (i.e. Randomization/Enrollment Date, Total Recruitment Goal, Expected Rate of Enrollment per Month), we would have one program to create this exact same slide for any number of studies which in turn provides efficiency and standardization across all presentations made to DSMB members. The following code provides the macro code and an example of two different macro calls. Figure 7 shows the two different slides that are produced from two individual macro calls.

```
%MACRO ENROLLMENTFIGURE(SUBDS=, RAND=, RANDDT=, GOAL=, RATEPM=, FOOTNOTE=);
/*Subject Level Dataset*/
PROC SQL NOPRINT;
    SELECT MIN(&RANDDT) INTO: FIRSTRAND FROM &SUBDS;
RUN;
```

```

DATA SUBS;
  SET &SUBDS;
  WHERE &RAND = 'Yes'; /*Randomization Indicator*/
  N = 1;
  MONTH = INTCK('MONTH', &FIRSTRAND, &RANDDT);
RUN;
PROC SORT DATA = SUBS;
  BY MONTH;
RUN;
PROC SQL NOPRINT;
  SELECT MAX(MONTH) INTO: LASTMONTH FROM SUBS;
RUN;

PROC FREQ DATA = SUBS NOPRINT;
  TABLES MONTH / OUT=MONTHS OUTCUM;
RUN;
DATA ACTUAL;
  SET MONTHS;
  GROUP = "Actual";
  RETAIN COUNT;
  SUBS=CUM_FREQ;
  KEEP MONTH SUBS GROUP;
RUN;
PROC SQL NOPRINT;
  SELECT PUT(MAX(SUBS), 3.) INTO: RANDN FROM ACTUAL;
  SELECT PUT(MAX(MONTH), 3.) INTO: LMON FROM ACTUAL;
RUN;
QUIT;
DATA PROJECT;
  GROUP = "Expected";
  PRO = 0;
  SUBSP=0;
  OUTPUT;
  PRO=(&GOAL./&RATEPM.);SUBSP=&GOAL.;OUTPUT;
RUN;
DATA PLOT1;
  SET ACTUAL PROJECT;
  LABEL GROUP = "Type: ";
RUN;
%MODSTYLE(NAME=STYLEONE, PARENT=JOURNAL, TYPE=CLM, COLORS=BLUE BLACK);
ODS POWERPOINT FILE = "&FILEPATH.\ENROLLMENTFIGURE.PPTX" NOGFOOTNOTE
STYLE=STYLEONE;
ODS GRAPHICS / WIDTH=8IN HEIGHT=8IN NOBORDER;
PROC SGPLOT DATA = PLOT1 NOAUTOLEGEND;
  SERIES X=PRO Y=SUBSP / LINEATTRS=(COLOR=BLACK) LEGENDLABEL="Expected"
NAME="e";
  STEP X=MONTH Y=SUBS / LINEATTRS=(COLOR=BLUE) CURVELABEL = "N= &RANDN. at
Month &LMON." CURVELABELATTRS=(SIZE=9PT) LEGENDLABEL="Actual" NAME="a";
  XAXIS LABEL = "Study Month" OFFSETMAX=0.25;
  YAXIS LABEL = "Number Randomized" OFFSETMAX=.05;
  KEYLEGEND "e" "a" / NOBORDER LOCATION=INSIDE POSITION=TOPLEFT;
  TITLE "Enrollment By Study Month" BOLD H=35PT;
  %IF &FOOTNOTE. ^= %THEN FOOTNOTE J=L "&FOOTNOTE";;
RUN;
ODS POWERPOINT CLOSE;
%MEND ENROLLMENTFIGURE;

```

```
%ENROLLMENTFIGURE (SUBDS=ADSL_100, RAND=RAND, RANDDT=RANDT, GOAL=100,
RATEPM=10);
```

```
%ENROLLMENTFIGURE (SUBDS=ADSL_300, RAND=RAND, RANDDT=RANDT, GOAL=300,
RATEPM=10);
```

Enrollment By Study Month Enrollment By Study Month




Figure 7. Enrollment over Time Slides Produced from Two Separate Macro Calls

FIGURE SLIDE HELPFUL HINT

If a figure that is created in a separate program is already sufficient for your presentation, or if a figure that will not be produced in SAS is needed in your presentation, you can still include the figure in your presentation. Some code to accomplish this can be found below:

```
/*Create a dummy dataset*/
DATA DUMMY;
  LABEL = "";
RUN;

/*Insert the figure using the PREIMAGE option in PROC REPORT*/
ODS LAYOUT GRIDDED COLUMNS=1;
ODS REGION WIDTH=7IN;
PROC REPORT DATA=DUMMY NOHEADER NOFS
  STYLE (REPORT)={PREIMAGE="<FIGURE PATH>"};
  COLUMN LABEL;
  DEFINE LABEL / NOPRINT ;
RUN;
ODS LAYOUT END;
```

The figure must be in a non-PDF picture format (e.g. TIFF, PNG, JPG, etc.). The file may require resizing prior to including in the presentation to fit in a slide, which can be done after opening the file in Microsoft Paint.

POPULATING A TEMPLATE

In order to facilitate standardization across all studies reporting to the DSMB, we created a template PowerPoint presentation. To create our DSMB presentation template, we saved all macros that create all necessary slides in a single folder, then created a program (let's call it DSMB_Slide_Macros.sas) that

used %INCLUDE statements to include each of the macros we needed. Here is a snippet of an example of the code to do this:

```
%INCLUDE "<Folder path>\Introduction_Slide_Macro.sas";
%INCLUDE "<Folder path>\Outline_Slide_Macro.sas";
%INCLUDE "<Folder path>\Study_Design_Slide_Macro.sas";
```

The program that holds all %INCLUDE statements was then included in each presentation program, followed by calls to each relevant macro bounded by an ODS POWERPOINT call. Using the above macros, some example code is shown here:

```
%INCLUDE "<Folder path>\DSMB_Slide_Macros.sas";
ODS POWERPOINT FILE="<Output Folder path>\DSMB_Presentation.sas";
  %Introduction_Slide(<Macro variable definitions>)
  %Outline_Slide(<Macro variable definitions>)
  %Study_Design_Slide(<Macro variable definitions>)
ODS POWERPOINT CLOSE;
```

Our template consists of 27 slides that we are using to generate DSMB presentations. Of those 27 slides, 19 are required while the other 8 are optional (as indicated in Table 1). Using the procedures that have been highlighted in the above examples, we created one program that creates all the slides needed for a DSMB presentation while facilitating certain levels of customization.

Slide in DSMB Template Presentation	Study 1 1st DSMB 1 Site, Food Allergy Actively Enrolling	Study 2 2nd DSMB 3 Sites, Hay Fever Actively Enrolling	Study 3 4th DSMB 6 Sites, Atopic Dermatitis No Longer Enrolling
Introduction	X	X	X
Outline	X	X	X
Study Design Text	X		
Study Design Figure	X		
DSMB Recommendations		X	X
Study Status	X	X	X
Enrollment Table	X	X	
Enrollment Table by Site - optional		X	
Enrollment Figure	X	X	
Demographics Table	X	X	X
Subject Status Flow Chart	X	X	X
Subject Status Table by Site - optional		X	X
Subject Status Visits Bar Chart - optional		X	
Reasons for Early Termination Table	X	X	X
Reasons for Early Termination Table by Site – optional		X	X
Study Stopping Rules	X	X	X
Study Stopping Rules Met	X	X	X
Serious Adverse Events Summary Table	X	X	X
New Serious Adverse Events		X	
New Serious Adverse Events Narrative - optional			X

Slide in DSMB Template Presentation	Study 1 1st DSMB 1 Site, Food Allergy Actively Enrolling	Study 2 2nd DSMB 3 Sites, Hay Fever Actively Enrolling	Study 3 4th DSMB 6 Sites, Atopic Dermatitis No Longer Enrolling
Non-Serious Adverse Event Summary Table	X	X	X
Treatment and Procedure Related Adverse Events	X	X	X
WAO Grading of Treatment and Procedure Related Systemic Reactions - optional		X	
Epinephrine Use Table – optional	X		
Protocol Deviation Table	X	X	X
Protocol Deviation Table by Site - optional		X	X
Appendices	X	X	X

Table 1. Example of slides required for various studies reporting to the same DSMB

The number of sites, therapeutic area, status of the study (i.e. enrolling, active with complete enrollment, closed to follow-up) and the number of times a study has been presented to the DSMB all play a role in determining the number of slides that need to be populated from the template.

Study 1 is a single site study that is presenting to the DSMB for the first time, is actively enrolling, and is in the Food Allergy therapeutic area. In contrast, Study 3 is a multi-site study that is presenting to the DSMB for the fourth time, is active but has completed enrollment, and is in the Atopic Dermatitis therapeutic area. These two studies are at very different points in the life cycle of a clinical trial and therefore provide the best opportunity to compare and contrast the macro calls that would be needed to populate our DSMB template. Table 2 shows 2 example programs containing macro calls that generate the set of slides needed for Study 1 and Study 3. For ease of comparison, we have removed the macro calls for the slides that both studies have in common.

Study 1 Program	Study 3 Program
<pre> OPTIONS NODATE; ODS POWERPOINT FILE="%REPDIR.\Study_1_DSMB_Report.pptx" NOGTITLE NOGFOOTNOTE; <Introduction> <Outline> %Studydesign(DESIGN = A randomized trial in food allergy , ARMS = %STR(Milk Consumption Milk Avoidance) , ENROLLGOAL = 100 randomized participants , PRIMARY = Proportion of subjects passing an oral food challenge at 52 weeks of age.) %EnrollmentTable(DSET = Study 1 Subject Level Dataset , LEVEL1LABEL = Screened , LEVEL1FL = SCREEN = 'Y' , LEVEL2LABEL = Consented , LEVEL2FL = Consent = 'Y' , LEVEL3LABEL = Randomized , LEVEL3FL = RAND = 'Y' , GOALS = %QUOTE(N/A 200 100)) %EnrollmentFigure(SUBDS = Study 1 Subject Level Dataset , RAND = Study 1 Randomization Indicator , RANDDT = Study 1 Randomization Date , GOAL = 100 , RATEPM = 4) <Demographic Table> <Subject Status Flow Chart> ... <Treatment and Procedure Related Adverse Events> %EpiUseSlide(EPIDS = Study 1 Epinephrine Use dataset , SUBDS = Study 1 Subject Level Dataset , IDVAR = Study 1 Subject Identifier , TRTAS = Study 1 Association with Treatment Flag , PROAS = Study 1 Association with Procedure Flag , AEAE = Study 1 Association with AE Flag , MOREONE = Study 1 More than One Epi Use Flag) %DevSlide(DEVDS = Study 1 Deviation Dataset , SUBDS = Study 1 Subject Level Dataset , IDVAR = Study 1 Subject Identifier , CAT = Study 1 Deviation category variable) ODS POWERPOINT CLOSE; </pre>	<pre> OPTIONS NODATE; ODS POWERPOINT FILE="%REPDIR.\Study_3_DSMB_Report.pptx" NOGTITLE NOGFOOTNOTE ; <Introduction> <Outline> %Studydesign(DESIGN = A randomized trial in atopic dermatitis , ARMS = %STR(Active Therapy Placebo) , ENROLLGOAL = 300 randomized participants , PRIMARY = EASI Score at 4 weeks) <Demographic Table> <Subject Status Flow Chart> %SubStatusBySite(DSET = Study 3 Subject Level Dataset , LEVEL1LABEL = Randomized , LEVEL1FL = RAND = 'Y' , LEVEL2LABEL = In Treatment Phase , LEVEL2FL = TREATP = 'Y' , LEVEL3LABEL = Completed Study , LEVEL3FL = COMP = 'Y' , LEVEL4LABEL = Terminated Early , LEVEL4FL = TERM = 'Y' , SITEVAR = Study 3 Site Variable) <Early Termination Table> ... <Serious Adverse Event Summary Table> %NewSaeText(SAE1ID = <ID of subject who had first SAE to display> , SAE1EVENT = Headache , SAE1TEXT = 13 year old boy experienced headache at school 4 hours after treatment. The school nurse called 911. At the emergency department of the hospital, the physician did not see any signs of a headache. SAE resolved, and per protocol, the participant was discontinued from study treatment.) <Non-Serious Adverse Event Summary Table> ... <Treatment and Procedure Related Adverse Events> %DevSlide(DEVDS = Study 3 Deviation Dataset , SUBDS = Study 3 Subject Level Dataset , IDVAR = Study 3 Subject Identifier , CAT = Study 3 Deviation category variable , CUTOFF = Study 3 Date of previous DSMB data cut , DEVDT = Study 3 Date of protocol deviation start variable , SITEVAR = Study 3 Site Variable) ODS POWERPOINT CLOSE; </pre>

Table 2. Sample code for creating 2 DSMB presentations for differing studies

CONCLUSION

Using ODS POWERPOINT to create your DSMB presentations helps eliminate many opportunities for errors due to hand-entering values or copying and pasting tables and figures into your slides. Furthermore, by creating standardized, yet customizable, macros to accommodate many studies reporting to a DSMB, it is possible to create efficiencies that will save time and ensure DSMB presentations have the same formatting for every presentation. This paper explored several of the methods and tips that we used to create our DSMB presentation macros and will give you a start on creating your own.

ACKNOWLEDGMENTS

This work was performed as part of the Statistical and Clinical Coordinating Center for the National Institute of Allergy and Infectious Diseases (NIAID), Division of Allergy, Immunology, and Transplantation (DAIT) of the National Institutes of Health (NIH) grant UM2AI117870.

RECOMMENDED READING

- Hunter, Tim. 2013. "A First Look at the ODS Destination for PowerPoint." *Proceedings of the SAS Global Forum 2013 Conference*. Available at <http://support.sas.com/resources/papers/proceedings13/041-2013.pdf>.
- Hunter, Tim. 2016. "A Second Look at the ODS Destination for PowerPoint." *Proceedings of the SAS Global Forum 2016 Conference*. Available at <http://support.sas.com/resources/papers/proceedings16/SAS3801-2016.pdf>.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Kaitie Lawson
Rho, Inc.
919-595-6313
Kaitie_Lawson@rhoworld.com

Brett Jepson
Rho, Inc.
919-595-6239
Brett_Jepson@rhoworld.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.